

9/10.4

Calculs financiers

Les rapports quotidiens de l'individu avec l'argent n'en font pas forcément un expert lorsqu'il s'agit de réaliser un placement fructueux, de gérer une épargne, de déjouer les pièges du crédit ou d'évaluer à terme la valeur résiduelle d'un patrimoine.

Les multiples possibilités de variation des paramètres permises par ce programme Basic aideront chacun à isoler la meilleure opportunité dans le cas d'un placement ou l'option la moins douloureuse s'il s'agit d'un emprunt. On peut ainsi constater, lors d'une demande de prêt personnel en vue d'un achat de voiture, que les sommes économisées simplement par le choix judicieux de la durée de l'emprunt peuvent représenter la valeur d'un CPC 6128.

ORGANISATION DU PROGRAMME

Le programme est très structuré et l'on peut se contenter de ne taper que les modules correspondants à son problème ; cependant, la première partie (lignes 1000 à 1470) initialise certaines variables, contient le menu général ainsi que deux petits sous-programmes d'usage général ; elle doit être disponible dans tous les cas. Il est recommandé de saisir les sous-programmes en les testant au fur et à mesure à l'aide des exemples donnés dans les recopies d'écran qui suivent chaque liste, les erreurs et la mise au point finale seront ainsi réduites à néant.

```

1000 REM -----CALCULS FINANCIERS -----
1010 P$="* Montant du capital initial (F
)
:"
1020 I$="* Taux d'intérêt nominal (%)
:"
1030 C$="* Nombre annuel de calculs de l
'intérêt:"
1040 R$="* Montant des versements (F)
:"
1050 N$="* Nombre de versements par an
:"
1060 T$="* Valeur future (F)
:"
1070 S$="* Montant des retraits (F)
:"
1080 PA$="* Prix d'achat (F)
:"
1090 TA$="* Taux de dépréciation (%)
:"
1100 VA$="* Prix de revente (F)
:"
1110 MODE 2
1120 GOTO 1230
1130 REM -----
1150 PRINT:PRINT"Autre calcul (O/N) :";
REP$=""
1160 REP$=INKEY$:IF REP$<>"O" AND REP$<>
"N" THEN 1160 ELSE RETURN
1170 REM -----
1190 INPUT"* Durée en années entières
:";YO:INPUT"* et mois
entiers :";M
1200 Y=(12*YO+M)/12:RETURN
1210 REM -----
1230 CLS:PRINT TAB(30) "CALCULS FINANCIERE
S.":PRINT
1240 PRINT" A - Valeur future d'un inves
tissement."
1250 PRINT" B - Valeur future d'une épar
gne."
1260 PRINT" C - Montant des versements n
écessaires à une épargne."
1270 PRINT" D - Valeur maximale des retr
aits sur un capital."
1280 PRINT" E - Montant initial d'un inv
estissement pour obtenir une valeur futu
re."
1290 PRINT" F - Montant du capital néces
saire à des retraits réguliers."
1300 PRINT" G - Valeur du taux d'intérêt
nominal et effectif sur un capital."
1310 PRINT" H - Echancier d'intérêts."
1320 PRINT" I - Montant du capital d'un
prêt."

```

```

1330 PRINT" J - Montant des remboursemen
ts d'un prêt."
1340 PRINT" K - Montant du dernier rembo
urserment."
1350 PRINT" L - Solde restant à payer su
r un prêt."
1360 PRINT" M - Durée d'un prêt."
1370 PRINT" N - Calcul du taux d'intérêt
nominal sur un prêt."
1380 PRINT" O - Tableau d'amortissement
d'un prêt."
1390 PRINT" P - Taux de dépréciation d'u
n bien matériel."
1400 PRINT" Q - Tableau de dépréciation
et de valeur résiduelle d'un bien."
1410 PRINT" R - Sortir du programme."
1420 PRINT:PRINT"* Question à traiter...
";
1430 M$=INKEY$:IF M$<"A" OR M$>"R" THEN
1430
1440 CLS: ON ASC(M$)-64 GOSUB 1490,1570,
1650,1730,1820,1900,1990,2080,2440,2530,
2620,2700,2780,2870,3050,3430,3520,1460
1450 GOTO 1230
1460 END
1470 REM -----

```

CALCULS FINANCIERS.

- A - Valeur future d'un investissement.
- B - Valeur future d'une épargne.
- C - Montant des versements nécessaires à une épargne.
- D - Valeur maximale des retraits sur un capital.
- E - Montant initial d'un investissement pour obtenir une valeur future.
- F - Montant du capital nécessaire à des retraits réguliers.
- G - Valeur du taux d'intérêt nominal et effectif sur un capital.
- H - Echancier d'intérêts.
- I - Montant du capital d'un prêt.
- J - Montant des remboursements d'un prêt.
- K - Montant du dernier remboursement.
- L - Solde restant à payer sur un prêt.
- M - Durée d'un prêt.
- N - Calcul du taux d'intérêt nominal sur un prêt.
- O - Tableau d'amortissement d'un prêt.
- P - Taux de dépréciation d'un bien matériel.
- Q - Tableau de dépréciation et de valeur résiduelle d'un bien.
- R - Sortir du programme.

* Question à traiter...

CHOIX A À H DU MENU : PLACER SON ARGENT**Investissement**

On dispose d'un capital fixe (revente d'un bien, héritage, gain au jeu, fruit d'une épargne...) qu'on place à un taux déterminé (banque, prêt à un particulier...) ou estimé (terres, immeubles, actions...). Menus A, E, G.

Épargne

C'est une part de l'argent économisé sur les dépenses qui constitue ou accroît un capital.

L'épargne peut être placée au fur et à mesure de sa constitution. Menus B et C.

Rente

Retraits réguliers sur un capital pendant une période donnée jusqu'à épuisement de ce dernier, ou retraits réguliers sur un capital sur une période infinie sans altération de ce dernier (qui est néanmoins victime de l'érosion monétaire dont on peut alors tenir compte pour éviter la baisse effective des revenus). Menus D et F.

```

1470 REM -----
1490 PRINT "* Valeur future d'un investi
ssement."
1500 PRINT:PRINT"Exemple : Quelle sera d
ans 10 ans la valeur d'un placement de 2
00000 F à 19% ?"
1510 PRINT:PRINT P$;:INPUT P:PRINT I$;:I
NPUT I:PRINT C$;:INPUT C:GOSUB 1190
1520 I=I/C/100:T=P*(1+I)^(C*Y)
1530 PRINT:PRINT"* Valeur future :";INT(
T*100+.5)/100;" F":GOSUB 1150
1540 IF REP$="O" THEN PRINT:GOTO 1510 EL
SE RETURN
1550 REM -----

```

* Valeur future d'un investissement.

Exemple : Quelle sera dans 10 ans la valeur d'un placement de 200000 F à 19% ?

```

* Montant du capital initial (F)      :? 200000
* Taux d'intérêt nominal (%)          :? 19
* Nombre annuel de calculs de l'intérêt: ? 4
* Durée en années entières           :? 10
*      et mois entiers                :? 0

```

* Valeur future : 1279946 F

Autre calcul (O/N) :

```

1550 REM -----
1570 PRINT"* Valeur future d'une série d
e dépôts réguliers (même montant à pério
des fixes)."
```

Exemple : Epargne obtenue au bout de 10 ans de versements mensuels d'un montant de 1000 F placés à 5%.

```

1580 PRINT:PRINT"Exemple : Epargne obten
ue au bout de 10 ans de versements mensu
els d'un montant de 1000 F placés à 5%."
1590 PRINT:PRINT R$;:INPUT R:PRINT I$;:I
NPUT I:PRINT N$;:INPUT N:I=I/N/100:GOSUB
1190
1600 T=R*((I+1)^(N*Y)-1)/I
1610 PRINT:PRINT"* Valeur future :";INT(
T*100+.5)/100;" F":GOSUB 1150
1620 IF REP$="O" THEN PRINT:GOTO 1590 EL
SE RETURN
1630 REM -----
```

* Valeur future d'une série de dépôts réguliers (même montant à périodes fixes).

Exemple : Epargne obtenue au bout de 10 ans de versements mensuels d'un montant de 1000 F placés à 5%.

| | |
|-------------------------------|---------|
| * Montant des versements (F) | :? 1000 |
| * Taux d'intérêt nominal (%) | :? 5 |
| * Nombre de versements par an | :? 12 |
| * Durée en années entières | :? 10 |
| * et mois entiers | :? 0 |

* Valeur future : 155286.1 F

Autre calcul (O/N) :

```

1630 REM -----
1650 PRINT"* Versements réguliers nécess
aires à l'obtention d'une épargne."
1660 PRINT:PRINT"Exemple : Combien doit
on placer chaque mois à 5.5% pour obteni
r 100000 F dans 5 ans ?"
1670 PRINT:PRINT T$;:INPUT T:PRINT I$;:I
NPUT I:PRINT N$;:INPUT N:I=I/N/100:GOSUB
 1190
1680 R=T*I/((I+1)^(N*Y)-1)
1690 PRINT:PRINT"* Montant des dépôts : "
;INT(R*100+.5)/100;" F":GOSUB 1150
1700 IF REP$="O" THEN PRINT:GOTO 1670 EL
SE RETURN
1710 REM -----

```

* Versements réguliers nécessaires à l'obtention d'une épargne.

Exemple : Combien doit on placer chaque mois à 5.5% pour obtenir 100000 F dans 5 ans ?

| | |
|-------------------------------|-----------|
| * Valeur future (F) | :? 100000 |
| * Taux d'intérêt nominal (%) | :? 5.5 |
| * Nombre de versements par an | :? 12 |
| * Durée en années entières | :? 5 |
| * et mois entiers | :? 0 |

* Montant des dépôts : 1451.78 F

Autre calcul (0/N) :

```

1710 REM -----
1730 PRINT"* Retraits maxima pouvant être effectués sur une période donnée."
1740 PRINT:PRINT"Exemple : Quel sera le montant maximum possible des retraits mensuels sur un capital de 80000 F placé à 15% pendant 5 ans, le solde résultant étant nul."
1750 PRINT:PRINT P$;:INPUT P:PRINT I$;:INPUT I
1760 PRINT"* Nombre de retraits par an :";:INPUT N:GOSUB 1190
1770 I=I/N/100:R=P*(I/((1+I)^(N*Y)-1)+I)
1780 PRINT:PRINT"* Montant maximum des retraits :";INT(R*100+.5)/100;" F":GOSUB 1150
1790 IF REP$="O" THEN PRINT:GOTO 1750 ELSE RETURN
1800 REM -----

```

* Retraits maxima pouvant être effectués sur une période donnée.

Exemple : Quel sera le montant maximum possible des retraits mensuels sur un capital de 80000 F placé à 15% pendant 5 ans, le solde résultant étant nul.

| | |
|----------------------------------|----------|
| * Montant du capital initial (F) | :? 80000 |
| * Taux d'intérêt nominal (%) | :? 15 |
| * Nombre de retraits par an | :? 12 |
| * Durée en années entières | :? 5 |
| * et mois entiers | :? 0 |

* Montant maximum des retraits : 1903.19 F

Autre calcul (O/N) :

```

1800 REM -----
1820 PRINT"* Investissement initial néce
ssaire à l'obtention d'une valeur future
."
1830 PRINT:PRINT"Exemple : Combien doit-
on placer à 7% pour avoir 5000 F au bout
de 2 ans ?"
1840 PRINT:PRINT T$;:INPUT T:PRINT I$;:I
NPUT I:PRINT C$;:INPUT C:I=I/C/100:GOSUB
1190
1850 P=T/(1+I)^(C*Y)
1860 PRINT:PRINT"* Investissement initia
l :";INT(P*100+.5)/100;" F":GOSUB 1150
1870 IF REP$="O" THEN PRINT:GOTO 1840 EL
SE RETURN
1880 REM -----

```

* Investissement initial nécessaire à l'obtention d'une valeur future.

Exemple : Combien doit-on placer à 7% pour avoir 5000 F au bout de 2 ans ?

```

* Valeur future (F)           :? 5000
* Taux d'intérêt nominal (%)  :? 7
* Nombre annuel de calculs de l'intérêt:? 12
* Durée en années entières    :? 2
*          et mois entiers     :? 0

```

* Investissement initial : 4348.56 F

Autre calcul (O/N) :


```

1880 REM -----
1900 PRINT"* Capital initial nécessaire
pour permettre des retraits réguliers pe
ndant un temps donné."
1910 PRINT:PRINT"Exemple : Combien doit-
on placer à 7% pour pouvoir retirer 1000
F par mois      durant 5 ans ?"
1920 PRINT:PRINT S$;:INPUT R:PRINT I$;:I
NPUT I:I=I/100
1930 PRINT"* Nombre de retraits par an
      :";:INPUT N:GOSUB 1190
1940 P=R*N/I*(1-1/((1+I/N)^(N*Y)))
1950 PRINT:PRINT"* Capital nécessaire : "
;INT(100*P+.5)/100;" F":GOSUB 1150
1960 IF REP$="O"THEN PRINT:GOTO 1920 ELS
E RETURN
1970 REM -----

```

* Capital initial nécessaire pour permettre des retraits réguliers pendant un temps donné.

Exemple : Combien doit-on placer à 7% pour pouvoir retirer 1000 F par mois durant 5 ans ?

| | |
|------------------------------|---------|
| * Montant des retraits (F) | :? 1000 |
| * Taux d'intérêt nominal (%) | :? 7 |
| * Nombre de retraits par an | :? 12 |
| * Durée en années entières | :? 5 |
| * et mois entiers | :? 0 |

* Capital nécessaire : 50502.13 F

Autre calcul (O/N) :

```

1970 REM -----
1990 PRINT"* Calcul des taux d'intérêts
nominal et effectif d'un investissement.
"
2000 PRINT:PRINT"Exemple : J'ai placé 5
00 F. Après 4 ans je possède 704 F. Quel
s sont les taux d'intérêt nominal et ef
fectifs ?"
2010 PRINT:PRINT P$;:INPUT P:PRINT T$;:I
NPUT T:PRINT"* Nombre annuel de calculs
de l'intérêt":PRINT" (tapez 0 en cas d'
inconnue)          :";:INPUT C:GOSUB 119
0
2020 IF C=0 THEN 2040
2030 Z=C*((T/P)^(1/(C*Y))-1)*100:PRINT:P
RINT"* Le taux nominal est ";Z;" %"
2040 PRINT"* Le taux effectif est ";((T/
P)^(1/Y)-1)*100;" %":GOSUB 1150
2050 IF REP$="O"THEN PRINT:GOTO 2010 ELS
E RETURN
2060 REM -----

```

* Calcul des taux d'intérêts nominal et effectif d'un investissement.

Exemple : J'ai placé 500 F. Après 4 ans je possède 704 F. Quels sont les taux d'intérêt nominal et effectifs ?

```

* Montant du capital initial (F)      :? 500
* Valeur future (F)                  :? 704
* Nombre annuel de calculs de l'intérêt
  (tapez 0 en cas d'inconnue)        :? 12
* Durée en années entières           :? 4
*      et mois entiers                :? 0

```

```

* Le taux nominal est : 8.584786 %
* Le taux effectif est 8.93079 %

```

Autre calcul (0/N) :

```

2060 REM -----
2080 PRINT"* Tableau des intérêts obtenu
s sur un capital."
2090 PRINT:PRINT P$;:INPUT P:PRINT I$;:I
NPUT I:I=I/100
2100 PRINT"* Nombre de dépôts ou retrait
s par an  ";:INPUT N1:IF N1=0 THEN 2130
2110 PRINT"* Montant de ceux-ci (avec +
ou -)      ";:INPUT R
2120 N=360:L2=N1:GOTO 2140
2130 PRINT C$;:INPUT N:N1=0:L2=4
2140 PRINT"* A quelle année commencer le
tableau  ";:INPUT X
2150 PRINT"* A quelle année le terminer
        ";:INPUT Y:X=INT(X):B0=P:I1=0
:I2=0:I3=0:K=66:P1=4
2160 FOR J0=1 TO INT(Y)+1
2170 IF J0<X THEN 2250
2180 IF K<55 THEN 2240
2190 K=6:PRINT:PRINT TAB(25)"Echéancier
d'intérêts"
2200 PRINT:PRINT P;"F à";I*100;"% nomina
l pour";Y;"ans. ";:IF N1=0 THEN 2220
2210 PRINT N1;"dépôts/retraits de";R;"F
par an.":K=K+1
2220 PRINT "Intérêt effectif";INT((100*(
(1+I/N)^N-1))*100+.5)/100;" % PAR AN"
2230 PRINT:PRINT"Année          Solde
Intérêts          intérêt cumul
és":PRINT
2240 PRINT J0;
2250 L1=1:N2=1:P2=1
2260 FOR J1=1 TO N:IF N2>N1 THEN 2290
2270 IF N2/N1>J1/N THEN 2290
2280 B0=B0+R:N2=N2+1
2290 B2=B0*(1+I/N):I1=B2-B0:I3=I3+I1:I2=
I2+I1
2300 IF P2/P1>J1/N THEN 2320
2310 P2=P2+1
2320 IF J0<X OR J1/N<L1/L2 THEN 2360
2330 L1=L1+1
2340 PRINT TAB(16) USING "#####.##"; B
2;:PRINT TAB(32) USING "#####.##";I3;:
PRINT TAB(48) USING "#####.##";I2
2350 I3=0:K=K+1
2360 B0=B2:IF J0+J1/N-1>=Y THEN J0= INT(
Y)+1:J1=N
2370 NEXT J1
2380 IF J0>=X THEN PRINT:K=K+1
2390 NEXT J0
2400 GOSUB 1150: IF REP$="O" THEN 2090
2410 RETURN
2420 REM -----

```

(suite)

* Tableau des intérêts obtenus sur un capital.

* Montant du capital initial (F) :? 100000
 * Taux d'intérêt nominal (%) :? 9.5
 * Nombre de dépôts ou retraits par an :? 4
 * Montant de ceux-ci (avec + ou -) :? -2400
 * A quelle année commencer le tableau :? 5
 * A quelle année le terminer :? 5

Echéancier d'intérêts

100000 F à 9.5 % nominal pour 5 ans. 4 dépôts/retraits de-2400 F par an.
 Intérêt effectif 9.96 % PAR AN

| Année | Solde | Intérêts | intérêt cumulés |
|-------|-----------|----------|-----------------|
| 5 | 100057.70 | 40857.70 | 40857.70 |
| | 100061.90 | 2404.20 | 43261.89 |
| | 100066.20 | 2404.27 | 45666.16 |
| | 100070.50 | 2404.37 | 48070.53 |

Autre calcul (O/N) :

* * *

* Tableau des intérêts obtenus sur un capital.

* Montant du capital initial (F) :? 10000
 * Taux d'intérêt nominal (%) :? 9.5
 * Nombre de dépôts ou retraits par an :? 0
 * Nombre annuel de calculs de l'intérêt: ? 12
 * A quelle année commencer le tableau :? 2
 * A quelle année le terminer :? 2

Echéancier d'intérêts

10000 F à 9.5 % nominal pour 2 ans. Intérêt effectif 9.92 % PAR AN

| Année | Solde | Intérêts | intérêt cumulés |
|-------|----------|----------|-----------------|
| 2 | 11255.62 | 1255.62 | 1255.62 |
| | 11525.07 | 269.44 | 1525.07 |
| | 11800.96 | 275.89 | 1800.96 |
| | 12083.46 | 282.50 | 2083.46 |

Autre calcul (O/N) :

CHOIX I à O DU MENU : EMPRUNTER DE L'ARGENT**Calcul impossible :**

Dans certains cas, alors que l'on est certain que le programme est correct, on peut obtenir un message d'erreur ou un dépassement de capacité ; ce problème apparaît du fait d'un calcul impossible comme c'est par exemple le cas si l'intérêt d'un emprunt s'accroît plus vite que son remboursement.

Valeur irréaliste

Il peut se produire que le dernier remboursement proposé dans le tableau d'amortissement soit négatif, les remboursements étaient trop élevés et la banque doit reverser de l'argent. Inversement, lorsque l'on rembourse un emprunt de 10 000 F à raison de 10 F par mois sur 5 ans, il est certain que le dernier remboursement sera supérieur au capital emprunté car chaque échéance ne couvre même pas le montant des intérêts. Il faut veiller à entrer des valeurs réalistes lors de l'exploitation et avant de demander un tableau, il est indispensable de calculer auparavant la durée du prêt ou la valeur exacte des remboursements.

Taux d'intérêts nominal et effectif

Le taux nominal est calculé une seule fois par an. Il rapporte légèrement moins que le taux effectif ou actuariel qui est composé plusieurs fois dans l'année (trimestriel, mensuel, bi-mensuel...) en tenant compte des intérêts passés dans le calcul des suivants. Menus G,H.

Prêt

Capital emprunté à une banque ou à un organisme de crédit. Le logiciel permet de constater que le prêt le plus intéressant se caractérise par son taux que l'on négociera évidemment au plus bas, sa durée, la plus courte possible, et le montant des remboursements, le plus élevé qu'on puisse supporter (en restant dans les limites tolérées par la loi). Menus I,J,M,O.

Remboursement anticipé

C'est une décision délicate à prendre car au début du remboursement la part d'amortissement du capital est faible par rapport à la part des intérêts (voir menu O). Le calcul du solde à payer à l'issue d'un temps donné (voir menu L) cause généralement de mauvaises surprises lorsque l'on s'aperçoit que l'on n'a pratiquement payé que des intérêts. De plus, le rachat d'un prêt entraîne généralement des frais supplémentaires imposés par la banque.

```

2420 REM -----
2440 PRINT"* Capital d'un prêt."
2450 PRINT:PRINT"Exemple : Combien peut-
on emprunter à 12% en remboursant mensue-
llement 900 F pendant 8 ans ?"
2460 PRINT:PRINT"* Valeur d'un rembourse-
ment :";:INPUT R:GOSUB 1190
2470 PRINT I$;:INPUT I:PRINT N$;:INPUT N
2480 P=R*N*(1-1/((I/100)/N+1)^(N*Y))/(I/
100)
2490 PRINT:PRINT"* Le capital est : "IN
T(P*100+.5)/100;"F":GOSUB 1150
2500 IF REP$="O" THEN PRINT:GOTO 2460 EL
SE RETURN
2510 REM -----

```

* Capital d'un prêt.

Exemple : Combien peut-on emprunter à 12% en remboursant mensuellement 900 F pendant 8 ans ?

```

* Valeur d'un remboursement           :? 900
* Durée en années entières           :? 8
*   et mois entiers                  :? 0
* Taux d'intérêt nominal (%)         :? 12
* Nombre de versements par an       :? 12

```

* Le capital est : 55374.93 F

Autre calcul (O/N) :

```

2510 REM -----
2530 PRINT"* Valeur du remboursement régulier permettant l'acquittement d'un prêt en un temps donné."
2540 PRINT:PRINT"Exemple : Quelle est la valeur du remboursement mensuel d'un prêt de 7000 F à 9% d'une durée de 2 ans et 6 mois ?"
2550 PRINT:GOSUB 1190
2560 PRINT P$;:INPUT P:PRINT I$;:INPUT I:PRINT N$;:INPUT N
2570 R=((I/100)*P/N)/(1-1/((I/100)/N+1)^(N*Y))
2580 PRINT:PRINT"* Valeur des remboursements :";INT(R*100+.5)/100;"F":GOSUB 1150
2590 IF REP$="O" THEN PRINT:GOTO 2550 ELSE RETURN
2600 REM -----

```

* Valeur du remboursement régulier permettant l'acquittement d'un prêt en un temps donné.

Exemple : Quelle est la valeur du remboursement mensuel d'un prêt de 7000 F à 9% d'une durée de 2 ans et 6 mois ?

* Durée en années entières :? 2
* et mois entiers :? 6
* Montant du capital initial (F) :? 7000
* Taux d'intérêt nominal (%) :? 9
* Nombre de versements par an :? 12

* Valeur des remboursements : 261.43 F

Autre calcul (O/N) :

```
2600 REM -----
2620 PRINT"* Coût du dernier rembourseme
nt d'un prêt."
2630 PRINT:PRINT R$;:INPUT R:PRINT P$;:I
NPUT P:PRINT I$;:INPUT I:PRINT N$;:INPUT
N:GOSUB 1190
2640 I=I/100
2650 FOR J1=1 TO N*Y:I1=INT((P*I/N)*100+
.5)/100:A=R-I1:P=P-A:NEXT J1
2660 PRINT:PRINT"* Montant du dernier ve
rsement :";INT((R+P)*100+.5)/100;"F":GOS
UB 1150
2670 IF REP$="O" THEN PRINT:GOTO 2630 EL
SE RETURN
2680 REM -----
```

* Coût du dernier remboursement d'un prêt.

| | |
|----------------------------------|-----------|
| * Montant des versements (F) | :? 12400 |
| * Montant du capital initial (F) | :? 200000 |
| * Taux d'intérêt nominal (%) | :? 5.5 |
| * Nombre de versements par an | :? 4 |
| * Durée en années entières | :? 4 |
| * et mois entiers | :? 7 |

* Montant du dernier versement : 16830.25 F

Autre calcul (0/N) :


```
2680 REM -----
2700 PRINT"* Solde restant dû sur un prê
t après un nombre de remboursements donn
és."
2710 PRINT:PRINT R$;:INPUT R:PRINT P$;:I
NPUT P:PRINT I$;:INPUT I:PRINT N$;:INPUT
N:I=I/100
2720 PRINT"* Nombre de remboursements ef
fectués  :";:INPUT N1
2730 FOR J=1 TO N1:I1=INT((P*I/N)*100+.5
)/100:A=R-I1:P=P-A:NEXT J
2740 PRINT:PRINT"* Solde restant dû :";I
NT(P*100+.5)/100;"F":GOSUB 1150
2750 IF REP$="O" THEN PRINT:GOTO 2710 EL
SE RETURN
2760 REM -----
```

* Solde restant dû sur un prêt après un nombre de remboursements donnés.

| | |
|--------------------------------------|---------|
| * Montant des versements (F) | :? 800 |
| * Montant du capital initial (F) | :? 9000 |
| * Taux d'intérêt nominal (%) | :? 25 |
| * Nombre de versements par an | :? 4 |
| * Nombre de remboursements effectués | :? 10 |

* Solde restant dû : 5832.58 F

Autre calcul (O/N) :

```

2760 REM -----
2780 PRINT"* Durée d'un prêt."
2790 PRINT:PRINT R$;:INPUT R:PRINT I$;:I
NPUT I
2800 PRINT"* Montant total du prêt
      :";:INPUT P:PRINT N$;:INPUT N
2810 Y=-(LOG(1-(P*(I/100))/(N*R))/(LOG(1
+I/100/N)*N))
2820 M=INT(Y*12+.5):Y0=INT(M/12):M=M-Y0*
12
2830 PRINT:PRINT"* Durée :";Y0;"ans";M;"
mois":GOSUB 1150
2840 IF REP$="O" THEN PRINT:GOTO 2790 EL
SE RETURN
2850 REM -----

```

* Durée d'un prêt.

| | |
|-------------------------------|-----------|
| * Montant des versements (F) | :? 12400 |
| * Taux d'intérêt nominal (%) | :? 5.5 |
| * Montant total du prêt | :? 200000 |
| * Nombre de versements par an | :? 4 |

* Durée : 4 ans 7 mois

Autre calcul (O/N) :

```

2850 REM -----
2870 PRINT"* Taux d'intérêt nominal chargé sur un prêt."
2880 PRINT:PRINT"Exemple : Quel est le taux d'intérêt d'un prêt de 9000 F d'une durée de 5 ans qui nécessite des remboursements de 800 F par trimestre ?"
2890 PRINT:PRINT R$;:INPUT R:GOSUB 1190
2900 PRINT"* Montant du capital
      :";:INPUT P
2910 PRINT N$;:INPUT N
2920 I=10:I2=0
2930 R1=(I*P/N)/(1-1/((I/N+1)^(N*Y)))
2940 R1=INT(R1*100+.5)/100
2950 I3=ABS(I-I2)/2:I2=I
2960 IF R1=R THEN 3000
2970 IF R1>R THEN 2990
2980 I=I+I3:GOTO 2930
2990 I=I-I3:GOTO 2930
3000 I=((INT((I*1000)*100+.5))/100)/1000
3010 PRINT:PRINT"* Taux d'intérêt nominal :";I*100;"%":GOSUB 1150
3020 IF REP$="O" THEN PRINT:GOTO 2890 ELSE RETURN
3030 REM -----

```

* Taux d'intérêt nominal chargé sur un prêt.

Exemple : Quel est le taux d'intérêt d'un prêt de 9000 F d'une durée de 5 ans qui nécessite des remboursements de 800 F par trimestre ?

| | |
|-------------------------------|---------|
| * Montant des versements (F) | :? 800 |
| * Durée en années entières | :? 5 |
| * et mois entiers | :? 0 |
| * Montant du capital | :? 9000 |
| * Nombre de versements par an | :? 4 |

* Taux d'intérêt nominal : 24.959 %

Autre calcul (O/N) :

```

3030 REM -----
3050 PRINT"* Tableau d'amortissement d'u
n prêt."
3060 PRINT:PRINT R$;:INPUT R:GOSUB 1190
3070 PRINT P$;:INPUT P:PRINT I$;:INPUT I
:I=I/100
3080 PRINT N$;:INPUT N
3090 PRINT"* Année de début
      :";:INPUT X:X=INT(X)
3100 N1=N:B0=P:C1=0:A1=0:I2=0:I3=0:A2=0
3110 IF INT(Y)>=1 THEN 3140
3120 N1=((Y-INT(Y))*12)/12*N
3130 J0=J0+1:GOTO 3150
3140 FOR J0=1 TO INT(Y)
3150 IF J0<X THEN 3200
3160 PRINT:PRINT TAB(25)"Tableau d'amort
issement."
3170 PRINT"Emprunt de";P;"F pour";YO;"an
(s) et";M;"mois à";I*100;"%";". Rembours
ements de";R;"F"
3180 PRINT "Année :";X
3190 PRINT" n°          part intérêt    p
art amortiss.      reliquat  Intérêts cumu
lés"
3200 FOR J1=1 TO N1:I1=INT((B0*I/N)*100+
.5)/100
3210 C1=C1+1:A=R-I1:A1=A1+A:B0=P-A1
3220 IF C1<>N*Y THEN 3240
3230 R=R+B0:A=A+B0:A1=A1+B0:B0=0
3240 I2=I2+I1:I3=I3+I1:A2=A2+A
3250 IF J0<X THEN 3280
3260 PRINT USING "##";J1;: PRINT TAB(16)
      USING "#####.##";I1;
3270 PRINT TAB(30) USING "#####.##"; (
INT(A*100+.5))/100;:PRINT TAB(44) USING
"#####.##";(INT(B0*100+.5))/100;:PRIN
T TAB(58) USING "#####.##";(INT(I2*100
+.5))/100
3280 NEXT J1
3290 IF C1<>N*Y THEN 3310
3300 PRINT"Le dernier paiement coûtera :
";(INT(R*100+.5))/100;"F"
3310 IF J0<X THEN 3370
3320 PRINT"Vous avez payé cette année : "
;

```

```

3330 PRINT I3;"F d'intérêts, et";
3340 PRINT (INT(A2*100+.5))/100;"F d'amortiss."
3350 PRINT"Appuyez sur RETURN pour la suite"
3360 K$=INKEY$:IF K$="" THEN 3360 ELSE X=X+1
3370 IF J0>=Y THEN 3400
3380 I3=0:A2=0:NEXT J0
3390 IF Y<>J0 THEN 3120
3400 GOSUB 1150: IF REP$="O" THEN PRINT:GOTO 3060 ELSE RETURN
3410 REM -----

```

* Tableau d'amortissement d'un prêt.

* Montant des versements (F) :? 16600
 * Durée en années entières :? 1
 * et mois entiers :? 0
 * Montant du capital initial (F) :? 60000
 * Taux d'intérêt nominal (%) :? 16.9
 * Nombre de versements par an :? 4
 * Année de début :? 1

Tableau d'amortissement.

Emprunt de 60000 F pour 1 an(s) et 0 mois à 16.9 %. Remboursements de 16600 F

Année : 1

| n° | part intérêt | part amortiss. | reliquat | Intérêts cumulés |
|----|--------------|----------------|----------|------------------|
| 1 | 2535.00 | 14065.00 | 45935.00 | 2535.00 |
| 2 | 1940.75 | 14659.25 | 31275.75 | 4475.75 |
| 3 | 1321.40 | 15278.60 | 15997.15 | 5797.15 |
| 4 | 675.88 | 15997.15 | 0.00 | 6473.03 |

Le dernier paiement coûtera : 16673.03 F

Vous avez payé cette année : 6473.03 F d'intérêts, et 60000 F d'amortiss.

Appuyez sur RETURN pour la suite

Autre calcul (0/N) :

MENUS P ET Q : DÉPRÉCIATION DES BIENS MATÉRIELS

Couplé à la consultation des tableaux d'argus des véhicules, le programme permet de déterminer quelles sont les marques ou les modèles d'automobiles auxquelles s'appliquent les moins mauvais taux de dépréciation et par conséquent ceux qui sont les plus cotés. On peut également fixer le juste prix de revente d'un bien en évitant les montants prohibitifs qui encombrant fréquemment les revues de petites annonces et qui n'attirent aucune proposition.

```

3410 REM -----
3430 PRINT "* Taux de dépréciation d'un
bien matériel."
3440 PRINT:PRINT "Exemple : J'ai acheté
une voiture 75000 F. Je la revends 42000
F deux ans et 3 mois plus tard."
3450 PRINT " - Quel a été son t
aux de dépréciation ?"
3460 PRINT:PRINT PA$;:INPUT P:PRINT VA$;
:INPUT V:GOSUB 1190
3470 T=100*(1-(V/P)^(1/Y))
3480 PRINT:PRINT TA$;INT(1000*T+.5)/1000
;"%":GOSUB 1150
3490 IF REP$="O" THEN PRINT: GOTO 3460 E
LSE RETURN
3500 REM -----

```

* Taux de dépréciation d'un bien matériel.

Exemple : J'ai acheté une voiture 75000 F. Je la revends 42000 F deux ans et 3 mois plus tard.

- Quel a été son taux de dépréciation ?

| | |
|----------------------------|------------|
| * Prix d'achat (F) | :? 75000 |
| * Prix de revente (F) | :? 42000 |
| * Durée en années entières | :? 2 |
| * et mois entiers | :? 3 |
| * Taux de dépréciation (%) | : 22.717 % |

Autre calcul (O/N) :

```

3500 REM -----
3520 PRINT "* Valeur résiduelle et monta
nt de la dépréciation d'un bien matériel
."
3530 PRINT:PRINT "Exemple : J'ai acheté
18000 F un micro-ordinateur. Ce modèle s
e déprécie à un  taux de 28% par an."
3540 PRINT"          - Combien vaut-il a
u bout d'un an ? De deux ans ? ..."
3550 PRINT"          - Combien perd-il l
a première année ? La deuxième ? ..."
3560 PRINT:PRINT PA$;:INPUT P:PRINT TA$;
:INPUT T
3570 PRINT:PRINT "Tableau sur 10 ans."
3580 PRINT"Année      Valeur résiduelle
Dépréciation"
3590 FOR I=1 TO 10
3600 PRINT USING "##";I;:PRINT TAB(15) U
SING "#####.##";INT(100*P*(1-T/100)^I
+.5)/100;:PRINT TAB(30) USING "#####.
##";INT(100*P*T/100*(1-T/100)^(I-1)+.5)/
100
3610 NEXT I:GOSUB 1150
3620 IF REP$="O" THEN PRINT:GOTO 3560 E
LSE RETURN
3630 REM -----

```

* Valeur résiduelle et montant de la dépréciation d'un bien matériel.

Exemple : J'ai acheté 18000 F un micro-ordinateur. Ce modèle se déprécie à un taux de 28% par an.

- Combien vaut-il au bout d'un an ? De deux ans ? ...
- Combien perd-il la première année ? La deuxième ? ...

* Prix d'achat (F) :? 18000
* Taux de dépréciation (%) :? 28

Tableau sur 10 ans.

| Année | Valeur résiduelle | Dépréciation |
|-------|-------------------|--------------|
| 1 | 12960.00 | 5040.00 |
| 2 | 9331.20 | 3628.80 |
| 3 | 6718.46 | 2612.74 |
| 4 | 4837.29 | 1881.17 |
| 5 | 3482.85 | 1354.44 |
| 6 | 2507.65 | 975.20 |
| 7 | 1805.51 | 702.14 |
| 8 | 1299.97 | 505.54 |
| 9 | 935.98 | 363.99 |
| 10 | 673.90 | 262.07 |

Autre calcul (O/N) :

